Rückmeldung von InputBox richtig auswerten
Versionen: Access 2007, 2003, 2002/XP, 2000 und 97

Um Informationen vom Anwender abzufragen, nutzen Sie in Ihren Datenbank-Anwendungen regelmäßig die Funktion "Inputbox()". Anschließend verwenden Sie eine Abfrage der folgenden Art, um festzustellen, ob der Anwender etwas eingegeben hat:

Dim strAnrede As String
Dim strEingabe As String
.....
strAnrede = Me.txtAnrede
strEingabe = Inputbox( _
             "Anrede:", _
             "Anrede eingeben/ändern:" , _
             strAnrede)
If strEingabe = "" Then Exit Sub
Me.txtAnrede = strEingabe

In diesem Beispiel soll vom Anwender eine Anrede wie "Herrn", "Frau" oder "Firma" eingegeben werden. Dazu wird der aktuelle Inhalt eines Feldes "txtAnrede" als Vorgabe gesetzt. Die Rückmeldung der Funktion "Inputbox()" ist die jeweilige Eingabe des Anwenders oder eine leere Zeichenfolge (""), wenn der Anwender auf Abbrechen geklickt hat. Was geschieht aber, wenn zum Beispiel der Anwender die aktuelle Anrede "Frau" löscht, weil er sich nicht ganz sicher ist, ob "Zadike Kalumbari" ein Männer- oder ein Frauenname ist? In diesem Fall würde die Abfrage

If strEingabe = "" Then Exit Sub

dazu führen, dass keine Änderung erfolgt, da die Eingabe vermeintlich per Klick auf Abbrechen bestätigt wurde. In einem solchen Fall wäre also zu klären, ob der Anwender wirklich eine leere Eingabe mit Ok bestätigt oder ob er auf Abrechen geklickt hat, um keine Änderung vorzunehmen:

Dim strAnrede As String
Dim strEingabe As String
Dim btn As Integer
.....
  strAnrede = Me.txtAnrede
  strEingabe = InputBox( _
               "Bitte Anrede eingeben:", _
               "Anrede eingeben/ändern:", _
               strAnrede)
  If strEingabe = "" Then
    btn = MsgBox(_
          "[OK] = Leere Eingabe zuweisen, " & _
          "[Abbrechen]= Keine Änderung?", _
          vbOKCancel + vbQuestion, _
          "Anrede eingeben/ändern:")
    If btn = vbOK Then
      Me.txtAnrede = vbNull
    Else
      Exit Sub
    End If
  Else
    Me.txtAnrede = strEingabe
  End If

Glücklicherweise ist ein solche Umstand nicht notwendig: Zur eindeutigen Beantwortung der Frage "Leere Eingabe oder "auf Abbrechen geklickt?" können Sie eine undokumentierte VBA-Funktion sowie ein spezielles Verhalten von "InputBox()" nutzen: Die Funktion

StrPtr(<Zeichenkettenvariable>)

liefert zunächst einen Pointer (Datentyp "Long Integer") auf eine Zeichenkette, der normalerweise für die Verwendung mit API-Funktionen genutzt wird. Ist die angegebene Zeichenkette nicht initialisiert, liefert "StrPtr()" das Ergebnis "0". "InputBox()" liefert bei einem Klick auf Abbrechen nicht wirklich eine leere Zeichenkette "", sondern eine nicht initialisierte Zeichenkette als Ergebnis. Daraus resultierend kann man also eine "InputBox()"-Rückmeldung wie folgt abfragen und eindeutig feststellen, ob eine leere, mit Ok bestätigte Eingabe vorliegt oder ob auf Abbrechen geklickt wurde:

Dim strAnrede As String
Dim strEingabe As String
.....
  strAnrede = Me.txtAnrede
  strEingabe = InputBox( _
               "Bitte Anrede eingeben:", _
               "Anrede eingeben/ändern:", _
               strAnrede)
  If StrPtr(strEingabe) = 0 Then
    MsgBox "Es wurde auf 'Abbrechen' geklickt!"
    Exit Sub
   ElseIf strEingabe = "" Then
    MsgBox "Eingabe wurde gelöscht!"
    Me.txtAnrede = vbNull
  Else
    MsgBox "Es wurde '" & _
           strEingabe & _
           "' eingegeben..."
    Me.txtAnrede = strEingabe
  End If

End Sub

Nach der Anzeige des Dialoges für die Eingabe wird zunächst mit

If StrPtr(strEingabe) = 0

auf Abbrechen geprüft. Trifft das nicht zu, prüft

ElseIf strEingabe = ""

auf eine leere Eingabe. Je nach Erfordernis kann dann davon abhängig zum Beispiel das Feld auf "vbNull" oder einen Standardwert zurückgesetzt werden.

